Chapter 18
Adding Angular Momenta in Quantum Mechanics

Let $\hat{\boldsymbol{J}}_1$ and $\hat{\boldsymbol{J}}_2$ be two angular momentum operators commuting with each other. Then the basis $|j_1, m_1; j_2, m_2>$ of common eigenstates of the operators $\hat{\boldsymbol{J}}_1^2$, $\hat{J}_{1z}$, $\hat{\boldsymbol{J}}_2^2$, $\hat{J}_{2z}$ exists. On the other hand, the total angular momentum $\hat{\boldsymbol{J}} = \hat{\boldsymbol{J}}_1 + \hat{\boldsymbol{J}}_2$ is also an angular momentum operator. Therefore, linear combinations $|j,m>$ of the states $|j_1, m_1; j_2, m_2>$ at given j_1, j_2 can be constructed in such a way that they are eigenstates of $\hat{\boldsymbol{J}}^2$ and $\hat{J}_z$. This problem [18] is called addition of the angular momenta j_1 and j_2.

We always have $m = m_1 + m_2$ because $\hat{J}_z = \hat{J}_{1z} + \hat{J}_{2z}$. The following figure illustrates addition of j_1 and j_2 (it assumes $j_1 \geqslant j_2$).

```
In[1] := Fig[j1_,j2_] := If[j1 < j2, Fig[j2,j1],
  With[{d = 0.75*j2, d2 = 0.1*j2, d3 = 0.15*j2, r = 0.05*j2},
    Graphics[Join[{Line[{{j2,j1+j2}, {j2,j2-j1}, {-j2,-j1-j2},
          {-j2,j1-j2}, {j2,j1+j2}}],
        Line[{{0,-j1-j2-d}, {0,j1+j2+d}}],
        Line[{{-j2-d,0}, {j2+d,0}}],
        Line[{{-d2,j1+j2}, {d2,j1+j2}}],
        Text[j1+j2, {-d3,j1+j2}, {1,0}],
        Line[{{-d2,-j1-j2}, {d2,-j1-j2}}],
        Text[-(j1+j2), {d3,-j1-j2}, {-1,0}],
        Line[{{-d2,j1-j2}, {d2,j1-j2}}],
        Text[j1-j2, {d3,j1-j2}, {-1,1}],
        Line[{{-d2,j2-j1}, {d2,j2-j1}}],
        Text[-(j1-j2), {-d3,j2-j1}, {1,-1}],
        Text[-j2, {-j2-d2,-d2}, {1,1}], Text[j2, {j2+d2,-d2}, {-1,1}],
        Text[m, {-d2,j1+j2+d}, {1,0}], Text[m2, {j2+d,-d2}, {0,1}]},
      Join[Table[{Disk[{m2,m},r]}, {m,j1-j2+1,j1+j2}, {m2,m-j1,j2}]],
      Join[Table[Disk[{m2,m},r], {m,-j1-j2,j2-j1-1}, {m2,-j2,m+j1}]],
      Join[Table[Disk[{m2,m},r], {m,j2-j1,j1-j2}, {m2,-j2,j2}]]]]]]
```


A. Grozin, *Introduction to Mathematica® for Physicists*, Graduate Texts in Physics,
DOI 10.1007/978-3-319-00894-3_18,

In[2] := Show[Fig[4, 2]]

Out[2] =



In[3] := Clear[Fig]

There is one state with $m = j_1 + j_2$, two states with $m = j_1 + j_2 - 1$, etc. Such an increase of the number of states occurs up to $m = j_1 - j_2$; further on it is constant up to $m = -(j_1 - j_2)$ and then decreases to one at $m = -(j_1 + j_2)$. Therefore, the maximum angular momentum resulting from adding j_1 and j_2 is $j = j_1 + j_2$. One state in the two-dimensional space of states with $m = j_1 + j_2 - 1$ refers to the same angular momentum, and the other one is the state with the maximum projection for the angular momentum $j = j_1 + j_2 - 1$. Continuing such reasoning, we see that all angular momenta up to $j_1 - j_2$ appear. In general, adding angular momenta j_1 and j_2 results in the angular momenta j from $|j_1 - j_2|$ to $j_1 + j_2$ in steps of 1.

This description naturally leads to the algorithm implemented below. We start from the only state with $m = j_1 + j_2$, namely the state $|j_1, j_1; j_2, j_2>$. It has $j = j_1 + j_2$, i.e., it is $|j_1 + j_2, j_1 + j_2>$. Repeatedly acting by the ladder operator $\hat{J}_- = \hat{J}_{1-} + \hat{J}_{2-}$ (and dividing by the appropriate normalization factor) we construct all the states with the total angular momentum $j = j_1 + j_2$: $|j_1 + j_2, j_1 + j_2 - 1>, \ldots,$ $|j_1 + j_2, -(j_1 + j_2)>$. Then we turn to the projection $m = j_1 + j_2 - 1$ and choose the state orthogonal to the already constructed one $|j_1 + j_2, j_1 + j_2 - 1>$. It has $j = j_1 + j_2 - 1$, i.e., it is $|j_1 + j_2 - 1, j_1 + j_2 - 1>$. Using the ladder operator we construct all the states with $j = j_1 + j_2 - 1$.

Then we proceed in a similar way. At the beginning of each step, when considering a new value of the projection m, we need to construct the state orthogonal to all the states with the same m already constructed. This is achieved as follows: we start from an arbitrary state, say, $|j_1, j_1; j_2, m - j_1>$, subtract its components along the already constructed states, and finally normalize the result. Then we construct all the states with the same total angular momentum from this state repeatedly acting by $\hat{J}_-$.

The function AddJ constructs the states $|j, m>$ (denoted Ket$[j, m]$) as linear combinations of the states $|j_1, m_1; j_2, m_2>$ (denoted ket[m1, m2]). It uses two local functions: Jm is the lowering operator $\hat{J}_-$ and ScaP is the scalar product. The procedure returns its local Ket, so that later the user will be able to inquire about Ket$[j, m]$ for specific values of j, m; in addition to this, the procedure prints all its results.

```
In[4] := AddJ = Function[{j1,j2},If[j2 > j1,AddJ[j2,j1],
  Module[{Ket, j, J, m,
      Jm = Function[{k},
        Expand[k/. ket[m1_,m2_]->
          Sqrt[(j1 - m1 + 1) * (j1 + m1)] * ket[m1 - 1,m2]+
            Sqrt[(j2 - m2 + 1) * (j2 + m2)] * ket[m1,m2 - 1]]],
      ScaP = Function[{k1,k2},
        Expand[k1 * k2]/.
          {ket[m1_,m2_]^2->1,ket[m1_,m2_] * ket[M1_,M2_]->0}]},
    Do[Ket[j, j] = ket[j - j2,j2];
      Do[Ket[j, j]- = Expand[ScaP[Ket[j, j],Ket[J, j]] * Ket[J, j]],
        {J,j1 + j2, j + 1, -1}];
      Print["Ket[", j, ",", j, "] = ",
        Ket[j, j] = Expand[Ket[j, j]/Sqrt[ScaP[Ket[j, j],Ket[j, j]]]]];
      Do[Print["Ket[", j, ",", m, "] = ",
          Ket[j, m] = Expand[Jm[Ket[j, m + 1]]/Sqrt[(j - m) * (j + m + 1)]]],
        {m, j - 1, -j, -1}],
     {j,j1 + j2,j1 - j2, -1}];
   Ket]]];
In[5] := AddJ[1/2, 1/2]
```

$$\mathrm{Ket}[1,1] = \mathrm{ket}\left[\frac{1}{2}, \frac{1}{2}\right]$$

$$\mathrm{Ket}[1,0] = \frac{\mathrm{ket}\left[-\frac{1}{2}, \frac{1}{2}\right]}{\sqrt{2}} + \frac{\mathrm{ket}\left[\frac{1}{2}, -\frac{1}{2}\right]}{\sqrt{2}}$$

$\text{Ket}[1,-1] = \text{ket}\left[-\frac{1}{2},-\frac{1}{2}\right]$

$\text{Ket}[0,0] = \frac{\text{ket}\left[-\frac{1}{2},\frac{1}{2}\right]}{\sqrt{2}} - \frac{\text{ket}\left[\frac{1}{2},-\frac{1}{2}\right]}{\sqrt{2}}$

Out[5] = Ket$668

In[6] := AddJ[1, 1/2]

$\text{Ket}\left[\frac{3}{2},\frac{3}{2}\right] = \text{ket}\left[1,\frac{1}{2}\right]$

$\text{Ket}\left[\frac{3}{2},\frac{1}{2}\right] = \sqrt{\frac{2}{3}}\text{ket}\left[0,\frac{1}{2}\right] + \frac{\text{ket}\left[1,-\frac{1}{2}\right]}{\sqrt{3}}$

$\text{Ket}\left[\frac{3}{2},-\frac{1}{2}\right] = \frac{\text{ket}\left[-1,\frac{1}{2}\right]}{\sqrt{3}} + \sqrt{\frac{2}{3}}\text{ket}\left[0,-\frac{1}{2}\right]$

$\text{Ket}\left[\frac{3}{2},-\frac{3}{2}\right] = \text{ket}\left[-1,-\frac{1}{2}\right]$

$\text{Ket}\left[\frac{1}{2},\frac{1}{2}\right] = \frac{\text{ket}\left[0,\frac{1}{2}\right]}{\sqrt{3}} - \sqrt{\frac{2}{3}}\text{ket}\left[1,-\frac{1}{2}\right]$

$\text{Ket}\left[\frac{1}{2},-\frac{1}{2}\right] = \sqrt{\frac{2}{3}}\text{ket}\left[-1,\frac{1}{2}\right] - \frac{\text{ket}\left[0,-\frac{1}{2}\right]}{\sqrt{3}}$

Out[6] = Ket$669

In[7] := AddJ[1, 1]

$\text{Ket}[2,2] = \text{ket}[1,1]$

$\text{Ket}[2,1] = \frac{\text{ket}[0,1]}{\sqrt{2}} + \frac{\text{ket}[1,0]}{\sqrt{2}}$

$\text{Ket}[2,0] = \frac{\text{ket}[-1,1]}{\sqrt{6}} + \sqrt{\frac{2}{3}}\text{ket}[0,0] + \frac{\text{ket}[1,-1]}{\sqrt{6}}$

$\text{Ket}[2,-1] = \frac{\text{ket}[-1,0]}{\sqrt{2}} + \frac{\text{ket}[0,-1]}{\sqrt{2}}$

$\text{Ket}[2,-2] = \text{ket}[-1,-1]$

$\text{Ket}[1,1] = \frac{\text{ket}[0,1]}{\sqrt{2}} - \frac{\text{ket}[1,0]}{\sqrt{2}}$

$\text{Ket}[1,0] = \frac{\text{ket}[-1,1]}{\sqrt{2}} - \frac{\text{ket}[1,-1]}{\sqrt{2}}$

$\text{Ket}[1,-1] = \frac{\text{ket}[-1,0]}{\sqrt{2}} - \frac{\text{ket}[0,-1]}{\sqrt{2}}$

$\text{Ket}[0,0] = \frac{\text{ket}[-1,1]}{\sqrt{3}} - \frac{\text{ket}[0,0]}{\sqrt{3}} + \frac{\text{ket}[1,-1]}{\sqrt{3}}$

Out[7] = Ket$670

In[8] := AddJ[2, 1]

$\text{Ket}[3,3] = \text{ket}[2,1]$

$\text{Ket}[3,2] = \sqrt{\frac{2}{3}}\text{ket}[1,1] + \frac{\text{ket}[2,0]}{\sqrt{3}}$

$\text{Ket}[3,1] = \sqrt{\frac{2}{5}}\text{ket}[0,1] + 2\sqrt{\frac{2}{15}}\text{ket}[1,0] + \frac{\text{ket}[2,-1]}{\sqrt{15}}$

$$\mathrm{Ket}[3,0]=\frac{\mathrm{ket}[-1,1]}{\sqrt{5}}+\sqrt{\frac{3}{5}}\mathrm{ket}[0,0]+\frac{\mathrm{ket}[1,-1]}{\sqrt{5}}$$
$$\mathrm{Ket}[3,-1]=\frac{\mathrm{ket}[-2,1]}{\sqrt{15}}+2\sqrt{\frac{2}{15}}\mathrm{ket}[-1,0]+\sqrt{\frac{2}{5}}\mathrm{ket}[0,-1]$$
$$\mathrm{Ket}[3,-2]=\frac{\mathrm{ket}[-2,0]}{\sqrt{3}}+\sqrt{\frac{2}{3}}\mathrm{ket}[-1,-1]$$
$$\mathrm{Ket}[3,-3]=\mathrm{ket}[-2,-1]$$
$$\mathrm{Ket}[2,2]=\frac{\mathrm{ket}[1,1]}{\sqrt{3}}-\sqrt{\frac{2}{3}}\mathrm{ket}[2,0]$$
$$\mathrm{Ket}[2,1]=\frac{\mathrm{ket}[0,1]}{\sqrt{2}}-\frac{\mathrm{ket}[1,0]}{\sqrt{6}}-\frac{\mathrm{ket}[2,-1]}{\sqrt{3}}$$
$$\mathrm{Ket}[2,0]=\frac{\mathrm{ket}[-1,1]}{\sqrt{2}}-\frac{\mathrm{ket}[1,-1]}{\sqrt{2}}$$
$$\mathrm{Ket}[2,-1]=\frac{\mathrm{ket}[-2,1]}{\sqrt{3}}+\frac{\mathrm{ket}[-1,0]}{\sqrt{6}}-\frac{\mathrm{ket}[0,-1]}{\sqrt{2}}$$
$$\mathrm{Ket}[2,-2]=\sqrt{\frac{2}{3}}\mathrm{ket}[-2,0]-\frac{\mathrm{ket}[-1,-1]}{\sqrt{3}}$$
$$\mathrm{Ket}[1,1]=\frac{\mathrm{ket}[0,1]}{\sqrt{10}}-\sqrt{\frac{3}{10}}\mathrm{ket}[1,0]+\sqrt{\frac{3}{5}}\mathrm{ket}[2,-1]$$
$$\mathrm{Ket}[1,0]=\sqrt{\frac{3}{10}}\mathrm{ket}[-1,1]-\sqrt{\frac{2}{5}}\mathrm{ket}[0,0]+\sqrt{\frac{3}{10}}\mathrm{ket}[1,-1]$$
$$\mathrm{Ket}[1,-1]=\sqrt{\frac{3}{5}}\mathrm{ket}[-2,1]-\sqrt{\frac{3}{10}}\mathrm{ket}[-1,0]+\frac{\mathrm{ket}[0,-1]}{\sqrt{10}}$$
$$\mathrm{Out}[8]=\mathrm{Ket\$671}$$